Biophysical Journal, Volume 97

Supporting Material

Impact of Emission Anisotropy on Fluorescence Spectroscopy and FRET Distance Measurements

Vassili Ivanov, Min Li, and Kiyoshi Mizuuchi

This Mathematica program was created and tested under Mathematica 7.0.1.0. It is also compatible with Mathematica 6. Execution of the whole program might take up to an hours on a computer with Core 2 Duo CPU. On Pentium 4 it might take many hours. Mathematica kernel might crash on Pentium 4 based computers during evaluation of this notebook because of CPU thermal design, not because of the bugs in Mathematica or this notebook.

Free *Mathematica* player can be downloaded from http://www.wolfram.com/products/player/

■ Definitions and Geometry

The X and Z axes are in the horizontal plane, the Y-axis is vertical.
We suggest that the illumination is cylindrically symmetric around the Z-axis.

These are unit vectors in the direction of the coordinate axes:

In[1]:= $\hat{k} = \{0, 0, 1\}; \; \hat{i} = \{1, 0, 0\}; \; \hat{j} = \{0, 1, 0\};$

Rotation operator:

In[2]:= $rop[\theta\_, \varphi\_] := \begin{pmatrix} Cos[\varphi] & -Sin[\varphi] & 0 \\ Sin[\varphi] & Cos[\varphi] & 0 \\ 0 & 0 & 1 \end{pmatrix} . \begin{pmatrix} Cos[\theta] & 0 & Sin[\theta] \\ 0 & 1 & 0 \\ -Sin[\theta] & 0 & Cos[\theta] \end{pmatrix}$

^The first rotation (right matrix) is rotation with the polar angle θ from Z-axis in the XZ plane;
the second rotation is in the XY plane (left matrix) with the azimuthal angle φ.

The second Legendre polynomial is

In[3]:= $p2[x\_] := \frac{3 x^2 - 1}{2}$

In[4]:= $iTrue = \frac{1}{3} (iPar + 2\, iPerp); \; iPerp = \frac{1 - r}{1 + 2 r} iPar;$

$iMeas = pA\, iPar + (1 - pA)\, iPerp; \; FullSimplify\left[\frac{iMeas - iTrue}{iTrue}\right]$

Out[4]= $(-1 + 3\, pA)\, r$

^ Eq. 2 is proved.

A. Effects of EA on the intensity measurements

Fractions of parallel illumination p

We start with illumination and detection profiles as two thin conical shells.
The result can be averaged over arbitrary cylindrically symmetric profies,
because it will be written as a bilinear function of the illumination and detector parameters.

Consider illumination and detection beams parallel near linear polarizers before focusing optics.
The index S will be used for the illumination light from the Source and the index D for the fluorescence going to Detector.
The ϕ is the angle between the illumination and detector optical axes.

Polarization of the illumination beam after the sourse polarizer, but before collimator is

```
In[5]:= polS = {Cos[ψS], Sin[ψS], 0}; polD = {Cos[ψD], Sin[ψD], 0};
```

where $\psi = 0$ for the horizontal orientation of a linear polarizer (H)
$\psi = \frac{\pi}{2}$ for the vertical orientation of a linear polarizer (V).

Polarization of illumination after the polarizer and after the collimator in the sample cuvette (with index C) is

```
In[6]:= polSC = FullSimplify[rop[0, ϕS].rop[αS, 0].rop[0, -ϕS].polS];
```

where αS is the angle of refraction in the illumination collimator,
ϕS is the angle around the collimator lens, which will be averaged for the cylindrical symmetry.

Direction of the illumination beam after the polarizer and after the collimator in the sample cuvette is

```
In[7]:= dirSC = FullSimplify[rop[0, ϕS].rop[αS, 0].rop[0, -ϕS].k̂];
```

We consider detector optics below, everything similar to the illumination optics pathway,
but we are tracking beams in the reverse direction from the detector back to the cuvette.
We also need extra rotation of the detector by the angle ϕ (the angle between the source and detector optical axes)
in the XZ plane around the vertical Y-axis using the operator rop[ϕ, 0].

Polarization of the fluorescence signal before polarizer and before the colimator in the cuvette for the fluorometer with arbitrary angle φ is

```
In[8]:= polDC = FullSimplify[rop[φ, 0].rop[0, ϕD].rop[αD, 0].rop[0, -ϕD].polD];
```

```
In[9]:= dirDC = FullSimplify[rop[φ, 0].rop[0, ϕD].rop[αD, 0].rop[0, -ϕD].k̂];
```

where αD is the angle of refraction, ϕD is the angle around the lens, which will be averaged for the cylindrical symmetry

The fraction of the parallel illumination p is the average $\cos^2$ of the angles between the illumination and detection polarizations in the sample chamber:

```
In[10]:= p = (FullSimplify[polSC.polDC])^2;
```

We need to average p over the angles αD and αS describing cylindrically symmetric shells in the illumination and detection profiles:

In[11]:=
```
tmp0 = Expand[p]; pAver = FullSimplify[
  1/(2 π)^2 Sum[Integrate[tmp0[[i]], {ϕD, 0, 2 π}, {ϕS, 0, 2 π}], {i, 1, Length[tmp0]}]];
```

$$\text{tmp0} = \text{Expand}[p];\ \text{pAver} = \text{FullSimplify}\left[\frac{1}{(2\pi)^2}\,\text{Sum}[\text{Integrate}[\text{tmp0}[[i]], \{\phi D, 0, 2\pi\}, \{\phi S, 0, 2\pi\}], \{i, 1, \text{Length}[\text{tmp0}]\}]\right];$$

Mathematica cannot efficiently simplify an equation on its own unless the result of simplification is zero.

Some guess work (intermediate steps are not shown) is required to find the simple form of p :

In[12]:=

$$\text{pAverGuess} = \frac{1}{6}\,(2 + \text{dD dS} + 3\,\text{bD bS}\,(\text{Cos}[2\,\psi D]\ \text{Cos}[2\,\psi S] + \text{Cos}[\varphi]\ \text{Sin}[2\,\psi D]\ \text{Sin}[2\,\psi S]) + \text{d}\varphi 1\,(\text{dD} + \text{bD Cos}[2\,\psi D])\,(\text{dS} + \text{bS Cos}[2\,\psi S]));$$

In[13]:=

$$\text{FullSimplify}\left[(\text{pAver} - \text{pAverGuess})\ /.\ \left\{\text{bD} \to \text{Cos}\left[\frac{\alpha D}{2}\right]^4,\ \text{bS} \to \text{Cos}\left[\frac{\alpha S}{2}\right]^4,\ \text{dD} \to \text{p2}[\text{Cos}[\alpha D]],\ \text{dS} \to \text{p2}[\text{Cos}[\alpha S]],\ \text{d}\varphi 1 \to (\text{p2}[\text{Cos}[\varphi]] - 1)\right\}\right]$$

Out[13]= 0

^ The result of our gesswork is equal to the unsimplified value of pAver. Eq. 3 is proved.

- **Calculating EA r from four polarized measurements**

In[14]:=
```
i[pA_, r_] := 1 + (-1 + 3 pA) r
```

^ We define intensity value up to some insignificatnt multiplier which cancells out in the calculation of EA.

In[15]:=

$$\text{solution} = \text{Solve}\left[\frac{(1 + \text{phh}\ r)\,(1 + \text{pvv}\ r)}{(1 + \text{phv}\ r)\,(1 + \text{pvh}\ r)} == \text{j2},\ r\right]$$

Out[15]=

$$\left\{\left\{r \to \frac{-\text{phh} + \text{j2 phv} + \text{j2 pvh} - \text{pvv} - \sqrt{(\text{phh} - \text{j2 phv} - \text{j2 pvh} + \text{pvv})^2 - 4\,(1 - \text{j2})\,(-\text{j2 phv pvh} + \text{phh pvv})}}{2\,(-\text{j2 phv pvh} + \text{phh pvv})}\right\},\right.$$
$$\left.\left\{r \to \frac{-\text{phh} + \text{j2 phv} + \text{j2 pvh} - \text{pvv} + \sqrt{(\text{phh} - \text{j2 phv} - \text{j2 pvh} + \text{pvv})^2 - 4\,(1 - \text{j2})\,(-\text{j2 phv pvh} + \text{phh pvv})}}{2\,(-\text{j2 phv pvh} + \text{phh pvv})}\right\}\right\}$$

In[16]:=
```
a1 = phh + pvv - j2 (phv + pvh);
a2 = j2 phv pvh - phh pvv;
```

In[18]:=

$$\text{FullSimplify}\left[\left\{\left(\frac{\text{a1} + \sqrt{\text{a1}^2 - 4\,(\text{j2} - 1)\,\text{a2}}}{2\,\text{a2}} - r\right)\ /.\ \text{solution}[[1]],\ \left(\frac{\text{a1} - \sqrt{\text{a1}^2 - 4\,(\text{j2} - 1)\,\text{a2}}}{2\,\text{a2}} - r\right)\ /.\ \text{solution}[[2]]\right\}\right]$$

Out[18]= {0, 0}

^ solution[[2]] is in the proper range, we chose it for Eq. 5

EXAMPLES: b- and d-factors for typical hardware configurations, Eqs. 14 and 15

The laser overfilling the aperture (a parallel beam with constant illumination intensity across the illumination collimator aperture), x=na/n:

In[19]:= $$\frac{\text{Integrate}\left[\text{p2}[\text{Cos}[\alpha]]\ \frac{\text{Sin}[\alpha]}{\text{Cos}[\alpha]^3},\ \{\alpha,\ 0,\ \text{ArcSin}[x]\}\right]}{\text{Integrate}\left[\frac{\text{Sin}[\alpha]}{\text{Cos}[\alpha]^3},\ \{\alpha,\ 0,\ \text{ArcSin}[x]\}\right]}$$

Out[19]= $$\text{If}\Big[(\text{ArcSin}[x] \notin \text{Reals} \ || \ 0 \le \text{Re}[\text{ArcCos}[x]] \le \pi) \ \&\& \ ((x \in \text{Reals} \ \&\& -1 \le \text{Re}[x] \le 1) \ || \ \text{Re}[x] == 0),\ \frac{1}{4}\left(\frac{x^2}{-1+x^2} - 3\,\text{Log}\left[1-x^2\right]\right),\ \text{Integrate}\left[\frac{3\,\text{Tan}[\alpha]}{2} - \frac{1}{2}\,\text{Sec}[\alpha]^2\,\text{Tan}[\alpha],\ \{\alpha,\ 0,\ \text{ArcSin}[x]\},\ \text{Assumptions} \to \ !\ ((\text{ArcSin}[x] \notin \text{Reals} \ || \ (\pi + 2\,\text{Re}[\text{ArcSin}[x]] \ge 0 \ \&\& \ 2\,\text{Re}[\text{ArcSin}[x]] \le \pi)) \ \&\& \ ((x \in \text{Reals} \ \&\& -1 \le \text{Re}[x] \le 1) \ || \ \text{Re}[x] == 0))\right]\Big] \Big/ \text{If}\left[\text{ArcSin}[x] \notin \text{Reals} \ || \ 0 \le \text{Re}[\text{ArcCos}[x]] \le \pi,\ \frac{x^2}{2-2\,x^2},\ \text{Integrate}\left[\text{Sec}[\alpha]^2\,\text{Tan}[\alpha],\ \{\alpha,\ 0,\ \text{ArcSin}[x]\},\ \text{Assumptions} \to \ !\ (\text{ArcSin}[x] \notin \text{Reals} \ || \ (\pi + 2\,\text{Re}[\text{ArcSin}[x]] \ge 0 \ \&\& \ 2\,\text{Re}[\text{ArcSin}[x]] \le \pi))\right]\right]$$

In[20]:= $$\text{FullSimplify}\left[\frac{\frac{1}{4}\left(\frac{x^2}{x^2-1} - 3\,\text{Log}\left[1-x^2\right]\right)}{\frac{x^2}{2-2\,x^2}}\right]$$

Out[20]= $$-\frac{1}{2} + \frac{3\left(-1+x^2\right)\text{Log}\left[1-x^2\right]}{2\,x^2}$$

In[21]:= $$\text{FullSimplify}\left[\frac{\frac{1}{4}\left(\frac{x^2}{x^2-1} - 3\,\text{Log}\left[1-x^2\right]\right)}{\frac{x^2}{2-2\,x^2}} - \text{p2}\left[\sqrt{\left(1-x^{-2}\right)\text{Log}\left[1-x^2\right]}\right]\right]$$

Out[21]= 0

In[22]:= $$\text{dSlaser}[x\_] := \text{p2}\left[\sqrt{\left(1-x^{-2}\right)\text{Log}\left[1-x^2\right]}\right]$$

In[23]:= $$\{\text{Limit}[\text{dSlaser}[x],\ x \to 0],\ \text{Limit}[\text{dSlaser}[x],\ x \to 1]\}$$

Out[23]= $$\left\{1,\ -\frac{1}{2}\right\}$$

In[24]:=
$$\frac{\text{Integrate}\left[\text{Cos}[\alpha/2]^4\,\frac{\text{Sin}[\alpha]}{\text{Cos}[\alpha]^3},\ \{\alpha,\ 0,\ \text{ArcSin}[x]\}\right]}{\text{Integrate}\left[\frac{\text{Sin}[\alpha]}{\text{Cos}[\alpha]^3},\ \{\alpha,\ 0,\ \text{ArcSin}[x]\}\right]}$$

Out[24]=
$$\text{If}\Big[(\text{ArcSin}[x]\notin\text{Reals}\ ||\ 0\le\text{Re}[\text{ArcCos}[x]]\le\pi)\ \&\&\ ((x\in\text{Reals}\ \&\&\ -1\le\text{Re}[x]\le 1)\ ||\ \text{Re}[x]==0),$$
$$\frac{1}{8}\left(-5+\frac{1}{1-x^2}+\frac{4}{\sqrt{1-x^2}}-\text{Log}\left[1-x^2\right]\right),\ \text{Integrate}\left[\text{Cos}\left[\frac{\alpha}{2}\right]^4\text{Sec}[\alpha]^2\,\text{Tan}[\alpha],\ \{\alpha,\ 0,\ \text{ArcSin}[x]\},$$
$$\text{Assumptions}\to\,!\,((\text{ArcSin}[x]\notin\text{Reals}\ ||\ (\pi+2\,\text{Re}[\text{ArcSin}[x]]\ge 0\ \&\&\ 2\,\text{Re}[\text{ArcSin}[x]]\le\pi))\ \&\&$$
$$((x\in\text{Reals}\ \&\&\ -1\le\text{Re}[x]\le 1)\ ||\ \text{Re}[x]==0))\Big]\Big]\Big/$$
$$\text{If}\left[\text{ArcSin}[x]\notin\text{Reals}\ ||\ 0\le\text{Re}[\text{ArcCos}[x]]\le\pi,\ \frac{x^2}{2-2x^2},\right.$$
$$\text{Integrate}\left[\text{Sec}[\alpha]^2\,\text{Tan}[\alpha],\ \{\alpha,\ 0,\ \text{ArcSin}[x]\},\right.$$
$$\left.\left.\text{Assumptions}\to\,!\,(\text{ArcSin}[x]\notin\text{Reals}\ ||\ (\pi+2\,\text{Re}[\text{ArcSin}[x]]\ge 0\ \&\&\ 2\,\text{Re}[\text{ArcSin}[x]]\le\pi))\right]\right]$$

In[25]:= **FullSimplify**[
$$\left(\frac{1}{8}\left(-5+\frac{1}{1-x^2}+\frac{4}{\sqrt{1-x^2}}-\text{Log}\left[1-x^2\right]\right)\Big/\frac{x^2}{2-2x^2}\right)-\left(\frac{5}{4}-x^{-2}\left(1-\sqrt{1-x^2}\right)+\frac{1}{4}\left(1-x^{-2}\right)\text{Log}\left[1-x^2\right]\right)\Big]$$

Out[25]= 0

In[26]:=
$$\text{bSlaser}[x\_]:=\frac{5}{4}-x^{-2}\left(1-\sqrt{1-x^2}\right)+\frac{1}{4}\left(1-x^{-2}\right)\text{Log}\left[1-x^2\right]$$

In[27]:= **{Limit[bSlaser[x], x → 0], Limit[bSlaser[x], x → 1]}**

Out[27]= $\left\{1,\ \frac{1}{4}\right\}$

The objective collecting light from an isotropic point-like source:

In[28]:=
$$\text{FullSimplify}\left[\frac{\text{Integrate}[\text{p2}[\text{Cos}[\alpha]]\,\text{Sin}[\alpha],\ \{\alpha,\ 0,\ \text{ArcSin}[x]\}]}{\text{Integrate}[\text{Sin}[\alpha],\ \{\alpha,\ 0,\ \text{ArcSin}[x]\}]}\right]$$

Out[28]=
$$\frac{1}{2}\left(1-x^2+\sqrt{1-x^2}\right)$$

In[29]:=
$$\text{dDobj}[x\_]:=\frac{1}{2}\left(1-x^2+\sqrt{1-x^2}\right)$$

In[30]:= **{Limit[dDobj[x], x → 0], Limit[dDobj[x], x → 1]}**

Out[30]= {1, 0}

In[31]:=
$$\text{FullSimplify}\left[\frac{\text{Integrate}\left[\text{Cos}[\alpha/2]^4\,\text{Sin}[\alpha],\ \{\alpha,\ 0,\ \text{ArcSin}[x]\}\right]}{\text{Integrate}[\text{Sin}[\alpha],\ \{\alpha,\ 0,\ \text{ArcSin}[x]\}]}\right]$$

Out[31]=
$$\frac{-22+15\sqrt{1-x^2}+6\,\text{Cos}[2\,\text{ArcSin}[x]]+\text{Cos}[3\,\text{ArcSin}[x]]}{48\left(-1+\sqrt{1-x^2}\right)}$$

In[32]:= $\mathtt{FullSimplify}\left[\frac{-22+15\sqrt{1-x^2}+6\,\mathrm{Cos}[2\,\mathrm{ArcSin}[x]]+\mathrm{Cos}[3\,\mathrm{ArcSin}[x]]}{48\left(-1+\sqrt{1-x^2}\right)}-\left(\frac{2}{3}-\frac{x^2}{12}+\frac{1}{3}\sqrt{1-x^2}\right)\right]$

Out[32]= 0

In[33]:= $\mathtt{bDobj}[x\_]:=\frac{2}{3}-\frac{x^2}{12}+\frac{1}{3}\sqrt{1-x^2}$

In[34]:= {Limit[bDobj[x], x → 0], Limit[bDobj[x], x → 1]}

Out[34]= $\left\{1,\ \frac{7}{12}\right\}$

- **G - factor**

Notations: g is g-factor, gG is G-factor.

In[35]:= $\mathtt{FullSimplify}\left[\mathtt{Solve}\left[\left\{\frac{ivh}{ihv}==\frac{g}{gG},\ \frac{ihh}{ivv}==\frac{\mathrm{Cos}[\phi]^2}{g\,gG}+\frac{\mathrm{Sin}[\phi]^2}{gG}\frac{ihv}{ivv}\right\},\ \{g,\ gG\}\right]\right]$

Out[35]= $\left\{\left\{gG\to\frac{ihv\,\mathrm{Sin}[\phi]^2-\frac{\sqrt{ihv}\sqrt{4\,ihh\,ivv\,\mathrm{Cos}[\phi]^2+ihv\,ivh\,\mathrm{Sin}[\phi]^4}}{\sqrt{ivh}}}{2\,ihh},\right.\right.$

$g\to\frac{ivh\,\mathrm{Sin}[\phi]^2-\frac{\sqrt{ivh}\sqrt{4\,ihh\,ivv\,\mathrm{Cos}[\phi]^2+ihv\,ivh\,\mathrm{Sin}[\phi]^4}}{\sqrt{ihv}}}{2\,ihh}\Big\},$

$\Big\{gG\to\frac{ihv\,\mathrm{Sin}[\phi]^2+\frac{\sqrt{ihv}\sqrt{4\,ihh\,ivv\,\mathrm{Cos}[\phi]^2+ihv\,ivh\,\mathrm{Sin}[\phi]^4}}{\sqrt{ivh}}}{2\,ihh},$

$\left.\left.g\to\frac{ivh\,\mathrm{Sin}[\phi]^2+\frac{\sqrt{ivh}\sqrt{4\,ihh\,ivv\,\mathrm{Cos}[\phi]^2+ihv\,ivh\,\mathrm{Sin}[\phi]^4}}{\sqrt{ihv}}}{2\,ihh}\right\}\right\}$

^ The proper solution has a plus sign before sqiare roots, Eqs. 7,8.

B. Estimate of the FRET DOF from the EA

- Averaging κ^2

In[36]:= $\mathtt{donor}=\mathtt{FullSimplify}\left[\mathtt{rop}[\theta_d,\ \phi_d].\mathtt{rop}[\theta 1,\ \phi 1].\hat{k}\right];$

The acceptor dipole:

In[37]:= $\mathtt{acceptor}=\mathtt{FullSimplify}\left[\mathtt{rop}[\theta_a,\ \phi_a].\mathtt{rop}[\theta 2,\ \phi 2].\hat{k}\right];$

The near field of donor in the location of acceptor is

In[38]:= $\mathtt{nearField}=\mathtt{FullSimplify}\left[3\,\hat{k}\left(\hat{k}.\mathtt{donor}\right)-\mathtt{donor}\right];$

The coupling efficiency κ^2 is

In[39]:=
```
k2 = (nearField.acceptor)^2;
```

In[40]:=
```
tmp0 = Expand[k2 * Sin[θ1] Sin[θ2]];
```

We are averaging over cylindrically symmetric distributions of the donor and acceptor dipoles by integrating over their azimuthal angles below:

In[41]:=
```
k2aver =
  FullSimplify[Sum[Integrate[tmp0[[i]], {ϕ1, 0, 2 π}, {ϕ2, 0, 2 π}], {i, 1, Length[tmp0]}] /
     Integrate[1 Sin[θ1] Sin[θ2], {ϕ1, 0, 2 π}, {ϕ2, 0, 2 π}]];
```

We skip algebraic simplification of k2α, which is straight forward. Reader can consider the result as a guess:

In[42]:=
```
k2Guess = (dA dD (2 Cos[θa] Cos[θd] - Cos[ϕa - ϕd] Sin[θa] Sin[θd])^2 + (1 - dA) (dD Cos[θa]^2 + 1/3) +
      (1 - dD) (dA Cos[θd]^2 + 1/3)) /. {dA → p2[Cos[θ1]], dD → p2[ Cos[θ2]]};
```

In[43]:=
```
FullSimplify[k2aver - k2Guess]
```

Out[43]= 0

^ The result of guess k2Guess is equal to calculation result k2aver.
Eqs. 21, 22 are proved for the angular distributions on the thin conical shells (θ1 and θ2 fixed).
Because the result is bilinear with respect to dA and dD,
the angular distribution can be arbitrary cylindrically symmetric as mentioned in the text op the paper.

- **Example: DOF uncertainty for HIV-1 integrase complex, Figs. 4 and 5**

In[44]:=
```
cosβ[rd_, ra_, rf_] := Sqrt[2/3 rf/Sqrt[rd ra] + 1/3];
```

In[45]:=
```
rd = 0.275; ra = 0.292; rf = -0.026;
```

Fig .4 Admissible area on (Θ_a, Θ_d) plane:

In[46]:=
```
ContourPlot[UnitStep[(Sin[θa] Sin[θd])^2 - (cosβ[rd, ra, rf] - Cos[θa] Cos[θd])^2 ],
  {θa, 0, π}, {θd, 0, π}, Contours → {1/2},
  ColorFunction → (If[#1 > 1/2, Black, White] &), BoundaryStyle → Black]
```

Out[46]=



These are Eqs. 21,22 with substitutted depolarization factors calculated using Eq. 19:

In[47]:=
$$\text{k2pm[rd\_, ra\_, rf\_]} := \left(1-\sqrt{\frac{5}{2}\,\text{ra}}\right)\left(\sqrt{\frac{5}{2}\,\text{rd}}\ \text{Cos}\left[\frac{\theta_{plus}-\theta_{minus}}{2}\right]^2+\frac{1}{3}\right)+\left(1-\sqrt{\frac{5}{2}\,\text{rd}}\right)\left(\sqrt{\frac{5}{2}\,\text{ra}}\ \text{Cos}\left[\frac{\theta_{plus}+\theta_{minus}}{2}\right]^2+\frac{1}{3}\right)+\frac{5}{2}\sqrt{\text{rd ra}}\left(\frac{3}{2}\left(\text{Cos}[\theta_{plus}]+\text{Cos}[\theta_{minus}]\right)-\text{cos}\beta[\text{rd, ra, rf}]\right)^2;$$

^ We use $\theta_{plus} = \theta_a + \theta_d$ and $\theta_{minus} = \theta_a - \theta_d$ for parametrization.

Fig. 5A Graph of $\kappa^2(\theta_{plus}, \theta_{minus})$ within the admissible area:

In[48]:=
$$\text{Plot3D}\left[\text{k2pm[rd, ra, rf]}, \{\theta_{minus}, -\text{ArcCos}[\text{cos}\beta[\text{rd, ra, rf}]], \text{ArcCos}[\text{cos}\beta[\text{rd, ra, rf}]]\}, \{\theta_{plus}, \text{ArcCos}[\text{cos}\beta[\text{rd, ra, rf}]], 2\pi-\text{ArcCos}[\text{cos}\beta[\text{rd, ra, rf}]]\}, \text{PlotLabel}\to\text{"A"}, \text{AxesLabel}\to\{\text{"}\theta_a\text{-}\theta_d\text{"}, \text{"}\theta_a\text{+}\theta_d\text{"}, \text{"}\kappa^2\text{"}\}\right]$$

Out[48]=



Fig. 5B Graph of $\triangle = \sqrt[6]{\frac{\kappa^2}{2/3}}\ (\theta_{plus}, \theta_{minus})$ within the admissible area:

In[49]:=

In[50]:= `Plot3D[` $\sqrt[6]{\frac{\text{k2pm[rd, ra, rf]}}{(2/3)}}$, `{θminus, -ArcCos[cosβ[rd, ra, rf]], ArcCos[cosβ[rd, ra, rf]]},`
`{θplus, ArcCos[cosβ[rd, ra, rf]], 2 π - ArcCos[cosβ[rd, ra, rf]]},`
`PlotLabel → "B", AxesLabel → {"θa-θd", "θa+θd", "Δ    "}]`

Out[50]=



- **Example: DOF uncertainty for HIV-1 integrase complex, r_{FRET} is unknown, Table 1, position 1**

r_{FRET} is unknown, use Eqs. 23 :

In[51]:= $\text{k2max[rd\_, ra\_]} := \left(1 - \sqrt{\frac{5}{2}\text{ ra}}\right)\left(\sqrt{\frac{5}{2}\text{ rd}} + \frac{1}{3}\right) + \left(1 - \sqrt{\frac{5}{2}\text{ rd}}\right)\left(\sqrt{\frac{5}{2}\text{ ra}} + \frac{1}{3}\right) + \frac{5}{2}\sqrt{\text{rd ra}}\ 4;$

In[52]:= $\text{k2min[rd\_, ra\_]} := \frac{1}{3}\left(\left(1 - \sqrt{\frac{5}{2}\text{ ra}}\right) + \left(1 - \sqrt{\frac{5}{2}\text{ rd}}\right)\right);$

In[53]:= $\left\{\text{k2min[rd, ra], k2max[rd, ra]}, \sqrt[6]{\frac{\text{k2min[rd, ra]}}{(2/3)}}, \sqrt[6]{\frac{\text{k2max[rd, ra]}}{(2/3)}}\right\}$

Out[53]= {0.105481, 3.2059, 0.735435, 1.29919}

- **Example: DOF uncertainty for HIV-1 integrase complex, r_{FRET} is known, Table 1, positions 1 and 2**

Position 1 :

r_d , r_a, and r_{FRET} are

In[54]:= `rd = 0.275; ra = 0.292; rf = -0.026;`

The angle between the donor and acceptor β_{da} (in degrees) is

In[55]:= $\frac{180}{\pi}$ ArcCos[cosβ[rd, ra, rf]]

Out[55]= 58.554

Minimum value of κ^2 from R.E. Dale, J. Eisinger, and W.E. Blumberg paper of 1979 (2), (admissible range is not taken into account):

In[56]:= k2min1979[rd_, ra_, rf_] :=

$$\frac{2}{3}\left(1-\left(\sqrt{\frac{5}{2}ra}+\sqrt{\frac{5}{2}rd}\right)\Big/ 2+\cos\beta[rd,ra,rf]\sqrt{\sqrt{\frac{5}{2}rd}\sqrt{\frac{5}{2}ra}\left(1-\sqrt{\frac{5}{2}rd}\right)\left(1-\sqrt{\frac{5}{2}ra}\right)}\right);$$

In[57]:= k2min1979[rd, ra, rf]

Out[57]= 0.15165

We are searching numerically for the minimum and maximum within the admissible range of angles:

In[58]:=

```
{FindMinimum[{k2pm[rd, ra, rf], -ArcCos[cosβ[rd, ra, rf]] ≤ θminus ≤ ArcCos[cosβ[rd, ra, rf]],
    ArcCos[cosβ[rd, ra, rf]] ≤ θplus ≤ (2 π - ArcCos[cosβ[rd, ra, rf]])},
   {θminus, 0.1}, {θplus, 0.1}],
 FindMaximum[{k2pm[rd, ra, rf], -ArcCos[cosβ[rd, ra, rf]] ≤ θminus ≤ ArcCos[cosβ[rd, ra, rf]],
    ArcCos[cosβ[rd, ra, rf]] ≤ θplus ≤ (2 π - ArcCos[cosβ[rd, ra, rf]])},
   {θminus, 0.1}, {θplus, 0.1}]}
```

Out[58]=

```
{{0.148887, {θminus → 0.0403949, θplus → 2.33648}},
 {2.50496, {θminus → -0.00282491, θplus → 1.02196}}}
```

^ Despite our search was restricted to admissible range of angles only,
we found smaller minimum 0.148887, than the minimum 0.15165 found using equation from the paper (2),
The equation for maximum is also wrong in (2).

The distance error range, -% and +%

In[59]:= $\left\{1-\sqrt[6]{\frac{0.14888662815277348`}{(2/3)}},\ \sqrt[6]{\frac{2.5049600953996665`}{(2/3)}}-1\right\}*100$

Out[59]= {22.1083, 24.6853}

Position 2 :

r_d , r_a, and r_{FRET} are

In[60]:= rd = 0.34; ra = 0.316; rf = 0.23;

The angle between the donor and acceptor β_{da} (in degrees) is

In[61]:= $\frac{180}{\pi}$ ArcCos[cosβ[rd, ra, rf]]

Out[61]= 26.4843

We are searching numerically for the minimum and maximum within the admissible range of angles:

In[62]:=
```
{FindMinimum[{k2pm[rd, ra, rf], -ArcCos[cosβ[rd, ra, rf]] ≤ θminus ≤ ArcCos[cosβ[rd, ra, rf]],
    ArcCos[cosβ[rd, ra, rf]] ≤ θplus ≤ (2 π - ArcCos[cosβ[rd, ra, rf]])},
   {θminus, 0.1}, {θplus, 0.1}],
 FindMaximum[{k2pm[rd, ra, rf], -ArcCos[cosβ[rd, ra, rf]] ≤ θminus ≤ ArcCos[cosβ[rd, ra, rf]],
    ArcCos[cosβ[rd, ra, rf]] ≤ θplus ≤ (2 π - ArcCos[cosβ[rd, ra, rf]])},
   {θminus, 0.1}, {θplus, 0.1}]}
```

Out[62]= $\{\{0.112428, \{\theta_{minus} \to -0.124452, \theta_{plus} \to 2.00251\}\}, \{3.334, \{\theta_{minus} \to 0.0015189, \theta_{plus} \to 0.462239\}\}\}$

The distance error range, -% and +%

In[63]:= $\left\{1 - \sqrt[6]{\frac{0.11242760345126897`}{(2/3)}}, \sqrt[6]{\frac{3.334002202378015`}{(2/3)}} - 1\right\} * 100$

Out[63]= {25.6706, 30.7704}

Conclusions, slow diffusion cases

Slow rotational diffusion, Figs. 8 and 7.

The FRET efficiencies for the slow rotating fluorophores are calculated for different distances by numeric averaging of the FRET efficiency over an ensemble of randomly oriented, but motionless FRET pairs.

In[64]:= `k2def[f1_, f2_, f3_] := (2 Cos[f1] Cos[f2] - Cos[f3] Sin[f1] Sin[f2])^2`

In[65]:= $\text{efficiency}[x\_, f1\_, f2\_, f3\_] := \frac{\text{k2def}[f1, f2, f3]}{\text{k2def}[f1, f2, f3] + \frac{2}{3} x^6}$

In[66]:= `efficiency[x, f1, f2, f3]`

Out[66]= $\frac{(2 \text{Cos}[f1] \text{Cos}[f2] - \text{Cos}[f3] \text{Sin}[f1] \text{Sin}[f2])^2}{\frac{2 x^6}{3} + (2 \text{Cos}[f1] \text{Cos}[f2] - \text{Cos}[f3] \text{Sin}[f1] \text{Sin}[f2])^2}$

In[67]:= $\text{eff}[x\_] := \frac{1}{2\pi} \text{NIntegrate}\Big[\text{efficiency}[x, \theta_a, \theta_d, \phi_{ad}] \text{Sin}[\theta_a] \text{Sin}[\theta_d], \left\{\theta_a, 0, \frac{\pi}{2}\right\}, \left\{\theta_d, 0, \frac{\pi}{2}\right\}, \{\phi_{ad}, 0, 2\pi\}, \text{Method} \to \{\text{GlobalAdaptive}, \text{MaxErrorIncreases} \to 10\,000\}, \text{MaxPoints} \to 10\,000\,000, \text{PrecisionGoal} \to 4\Big]$

In[68]:=
```
fTable = Table[{x, eff[x]}, {x, 1 / 10, 2, 1 / 10}];
```

NIntegrate::slwcon :
Numerical integration converging too slowly; suspect one of the following: singularity, value of the integration is 0, highly oscillatory integrand, or WorkingPrecision too small. ≫

NIntegrate::slwcon :
Numerical integration converging too slowly; suspect one of the following: singularity, value of the integration is 0, highly oscillatory integrand, or WorkingPrecision too small. ≫

NIntegrate::slwcon :
Numerical integration converging too slowly; suspect one of the following: singularity, value of the integration is 0, highly oscillatory integrand, or WorkingPrecision too small. ≫

General::stop : Further output of NIntegrate::slwcon will be suppressed during this calculation. ≫

^ Because the integral is converging slowly *Mathematica* produces warnings.

In[69]:=
```
efR = Interpolation[fTable];
```

In[70]:= $\text{Plot}\left[\left\{\text{efR}[x], \frac{1}{1+x^6}\right\}, \{x, 0.1, 2\}, \text{AxesOrigin} \to \{0, 0\}, \text{AxesLabel} \to \left\{"\frac{R}{R_0}", "E"\right\},\right.$
$\text{PlotStyle} \to \{\{\text{Black}, \text{Thickness}[0.003]\}, \{\text{Black}, \text{Thickness}[0.003], \text{Dashed}\}\},$
$\left.\text{AxesStyle} \to \text{Directive}[\text{Black}, 14]\right]$

Out[70]=


^ This is Fig. 8. Dependence of the FRET efficiency on distance between fluorophores for the fast rotating fluorophores (dashed line) and the slow rotating fluorophores (solid line).

The effective ensemble average DOF for the slow rotating fluorophores (solid line) is calculated from the FRET efficiency curve (Fig. 8):

In[71]:= $\text{Solve}\left[\frac{1}{1+\frac{2}{3}x^6/\text{k2eff}} == \text{eff}, \text{k2eff}\right]$

Out[71]= $\left\{\left\{\text{k2eff} \to -\frac{2\,\text{eff}\,x^6}{3\,(-1+\text{eff})}\right\}\right\}$

In[72]:= $\text{k2eff}[x\_] := -\frac{2\,\text{efR}[x]\,x^6}{3\,(-1+\text{efR}[x])}$

In[73]:=
```
Plot[{k2eff[x], 2 / 3}, {x, 0.15, 2}, AxesOrigin → {0, 0}, AxesLabel → {"R/R0", "κ^2"},
  PlotStyle → {{Black, Thickness[0.003]}, {Black, Thickness[0.003], Dashed}},
  AxesStyle → Directive[Black, 14]]
```

Out[73]=


^ This is Fig. 7. Dependence of the DOF on the distance between fluorophores if the fluorophores are rotating without constraints. The DOF for the fast rotating fluorophores (dashed line) is equal 2/3 .

- **Any polarization due to slow angular diffusion? Fig. 9**

The EA for each distance was calculated by numeric averaging of $\frac{2}{5}$ p2[cosβ12[θ_a, θ_d, ϕ_{ad}] over anisotropic angular distribution with the extra weight equal to the FRET efficiency, and normalized by the average FRET efficiency calculated above.

In[74]:=
```
cosβ12[f1_, f2_, f3_] := Cos[f1] Cos[f2] + Cos[f3] Sin[f1] Sin[f2]
```

In[75]:=
```
efficiency[x_, f1_, f2_, f3_] := 1 / (1 + 2/3 x^6 / k2def[f1, f2, f3])
```

In[76]:=
```
ep2[x_] := 1/(8 π) NIntegrate[ p2[cosβ12[θa, θd, ϕad]] efficiency[x, θa, θd, ϕad] Sin[θa] Sin[θd],
   {θa, 0, π/2}, {θd, 0, π/2}, {ϕad, 0, 2 π}, Method → {GlobalAdaptive, MaxErrorIncreases → 10 000},
   MaxPoints → 10 000 000, PrecisionGoal → 4]
```

In[77]:=
```
eff[x_] := 1/(8 π) NIntegrate[ efficiency[x, θa, θd, ϕad] Sin[θa] Sin[θd], {θa, 0, π/2},
   {θd, 0, π/2}, {ϕad, 0, 2 π}, Method → {GlobalAdaptive, MaxErrorIncreases → 10 000},
   MaxPoints → 10 000 000, PrecisionGoal → 4]
```

In[78]:=
```
rAve[x_] := 2/5 ep2[x]/eff[x]
```

In[79]:=
```
rTable = Table[{x, rAve[x]}, {x, 0.2, 2, 0.2}];
```

NIntegrate::slwcon :
Numerical integration converging too slowly; suspect one of the following: singularity, value of the integration is 0, highly oscillatory integrand, or WorkingPrecision too small. ≫

NIntegrate::eincr :
The global error of the strategy GlobalAdaptive has increased more than 10000 times. The global error is expected to decrease monotonically after a number of integrand evaluations. Suspect one of the following: the working precision is insufficient for the specified precision goal; the integrand is highly oscillatory or it is not a (piecewise) smooth function; or the true value of the integral is 0. Increasing the value of the GlobalAdaptive option MaxErrorIncreases might lead to a convergent numerical integration. NIntegrate obtained 0.0040596193467080256` and 0.00015499174687725462` for the integral and error estimates. ≫

NIntegrate::slwcon :
Numerical integration converging too slowly; suspect one of the following: singularity, value of the integration is 0, highly oscillatory integrand, or WorkingPrecision too small. ≫

NIntegrate::slwcon :
Numerical integration converging too slowly; suspect one of the following: singularity, value of the integration is 0, highly oscillatory integrand, or WorkingPrecision too small. ≫

General::stop : Further output of NIntegrate::slwcon will be suppressed during this calculation. ≫

^ Because the integration is converging slowly and the accuracy of numeric calculation is not very high *Mathematica* produces warnings.

In[80]:=
```
ListLinePlot[rTable, AxesOrigin → {0, 0}, AxesLabel → {"R/R0", "r"},
  PlotStyle → {Black, Thickness[0.003]}, AxesStyle → Directive[Black, 14], Mesh → All]
```

Out[80]=



^ This is Fig. 9. EA of the FRET signal from the slow rotating fluorophore pair as a function of the distance between fluorophores. The numeric result is not very precise due to the high dimension of integration.

■ **Distance averaging: Slow and fast diffusion, Fig. 10**

The FRET efficiency was numerically calculated for the uniform distance distributions over +/- 20% and +/- 50% of R_0 by averaging FRET efficiency over distance distribution for the slow distance diffusion case and by averaging of the kinetic rate and calculating FRET efficiency from it for the fast distance diffusion.

In[81]:=
```
ee[x_] := 1/(1 + x^6)
```

In[82]:=
```
eSlow[x_, deltax_] := NIntegrate[ee[x + dx], {dx, -deltax, deltax}] / (2 deltax)
```

In[83]:=
```
kFast[x0_, dx_] := NIntegrate[x^6, {x, x0 - dx, x0 + dx}] / (2 dx)
```

In[84]:=
```
Plot[{ee[x], eSlow[x, 0.2], eSlow[x, 0.5], 1/(1 + kFast[x, 0.2]), 1/(1 + kFast[x, 0.5])},
  {x, 0, 2}, AxesOrigin → {0, 0}, AxesLabel → {"R/R0", "E"},
  PlotStyle → {{Black, Thickness[0.003]}, {Black, Thickness[0.003], Dashed},
    {Black, Thickness[0.003], Dashed}, {Black, Thickness[0.003], Dotted},
    {Black, Thickness[0.003], Dotted}}, AxesStyle → Directive[Black, 14]]
```

Out[84]=



^ This is Fig. 10, Dependence of the average FRET efficiency on the average distance between fluorophores for the different ranges of the distance diffusion for the molecules with fast rotational diffusion and fixed distance (solid line), with fast rotational diffusion and fast distance diffusion (dashed lines), fast rotational diffusion and slow distance diffusion (dotted lines). Smaller width of the distance distribution corresponds to the smaller deviation from the case with fast rotational diffusion and fixed distance.